

Gramática Libre de Contexto (GLC) – JavaUSAC

202200147 Mario Rodrigo Balam Churunel.

Entrega de Gramatica F1 compiladores.

Producción Inicial

INICIO ::= INSTRUCCIONES ;

INSTRUCCIONES ::= INSTRUCCIONES INSTRUCCION
| INSTRUCCION ;

Instrucciones Generales

INSTRUCCION ::= PRINT PAR1 EXPRESION PAR2 FINCADENA

| DECLARACION

| ASIGNACION

| SIF

| WHILE

| DO_WHILE

| FOR

| SWITCH_STM

| BREAK

| CONTINUE

| ID MASMAS FINCADENA

| ID MENOSMENOS FINCADENA

| error FINCADENA

| error

;

Sentencias de Control (If, While, Do-While)

WHILE ::= _while PAR1 EXPRESION PAR2 _llaveizq INSTRUCCIONES _llaveder
;

DO_WHILE ::= _do _llaveizq INSTRUCCIONES _llaveder _while PAR1 EXPRESION PAR2
FINCADENA
;

SIF ::= _if PAR1 EXPRESION PAR2 _llaveizq INSTRUCCIONES _llaveder
| _if PAR1 EXPRESION PAR2 _llaveizq INSTRUCCIONES _llaveder _else _llaveizq
INSTRUCCIONES _llaveder
| _if PAR1 EXPRESION PAR2 _llaveizq INSTRUCCIONES _llaveder ELSEIF
| _if PAR1 EXPRESION PAR2 _llaveizq INSTRUCCIONES _llaveder ELSEIF _else _llaveizq
INSTRUCCIONES _llaveder
;

ELSEIF ::= ELSEIF _else _if PAR1 EXPRESION PAR2 _llaveizq INSTRUCCIONES _llaveder
| _else _if PAR1 EXPRESION PAR2 _llaveizq INSTRUCCIONES _llaveder
;

Sentencia Switch

SWITCH_STM ::= _switch PAR1 EXPRESION PAR2 _llaveizq CASES_LIST DEFAULT_STM
_llaveder
| _switch PAR1 EXPRESION PAR2 _llaveizq CASES_LIST _llaveder
| _switch PAR1 EXPRESION PAR2 _llaveizq DEFAULT_STM _llaveder
;

CASES_LIST ::= CASES_LIST CASE_STM
| CASE_STM
;

CASE_STM ::= _case EXPRESION DOSPUNTOS INSTRUCCIONES

;

DEFAULT_STM ::= _default DOSPUNTOS INSTRUCCIONES

;

Sentencia For

FOR ::= _for PAR1 ASIGNACION_FOR EXPRESION FINCADENA ACTUALIZAR_FOR PAR2
_llaveizq INSTRUCCIONES _llaveder

;

ASIGNACION_FOR ::= DECLARACION

| ASIGNACION

;

ACTUALIZAR_FOR ::= ID IGUAL EXPRESION

| ID MASMAS

| ID MENOSMENOS

;

Sentencias de Transferencia

BREAK ::= _break FINCADENA

;

CONTINUE ::= _continue FINCADENA

;

Declaración y Asignación

DECLARACION ::= VAR ID DOSPUNTOS TIPOS IGUAL EXPRESION FINCADENA
| VAR ID DOSPUNTOS TIPOS FINCADENA
;

ASIGNACION ::= ID IGUAL EXPRESION FINCADENA
;

TIPOS ::= INT
| DOUBLE
| STRING
| BOOL
| CHAR_TYPE
;

Expresiones (Aritméticas, Lógicas, Relacionales)

EXPRESION ::= MENOS EXPRESION
| EXPRESION MAS EXPRESION
| EXPRESION MENOS EXPRESION
| EXPRESION MULTIPLICACION EXPRESION
| EXPRESION DIVISION EXPRESION
| EXPRESION POTENCIA EXPRESION
| EXPRESION MODULO EXPRESION
| EXPRESION EQUALS EXPRESION
| EXPRESION DIFERENTE EXPRESION
| EXPRESION _menorq EXPRESION
| EXPRESION _menorigual EXPRESION
| EXPRESION _mayorq EXPRESION
| EXPRESION _mayorigual EXPRESION
| EXPRESION AND EXPRESION

| EXPRESION OR EXPRESION
| EXPRESION XOR EXPRESION
| NOT EXPRESION
| PAR1 TIPOS PAR2 EXPRESION
| PAR1 EXPRESION PAR2
| ENTERO
| DECIMAL
| CHAR
| CADENA
| TRUE
| FALSE
| ID

;